
概 要

複雑な数式の変形や，外形のプロットなどは理論から計算へ結びつける際の基本となる．こ
のような骨の折れる作業を最近ではパソコンで瞬時に実行してくれるソフトが開発されている．
材料研究でよく出会う題材を使って，基本となる操作の習得を目的とする．原子レベルのシミュ
レーションをおこなう際には，そこで使われる原子間ポテンシャルがどの程度の精度かを認識し
ておく必要がある．原子間の相互作用を表現する一般的な原子間ポテンシャルを取り上げて，そ
の背景にある物理的考察と適用限界について詳述する．1

1 はじめに
原子レベルの理論計算では，電子構造を求める精度の高い第一原理計算に席巻されたかの様

相を呈している．しかし，ナノメータオーダーの現実的な現象をモデル化するには，いまだに多
くの重要な知見が，経験的な原子間ポテンシャルを用いた大規模シミュレーションによって得ら
れている．これは単に第一原理計算につきまとう計算機の性能や数値計算の誤差だけの問題では
なく，どれだけ現象のメカニズムにたいして直観的な理解が得られるかという描像の問題でもあ
る．現在の計算機の処理能力と，我々材料研究者の処理能力から経験ポテンシャルの利用は不可
欠である．
この第一原理計算による電子構造の理論的知識と，経験ポテンシャルで使われている原子間

結合の経験的知識とをつなぐことがこの講義の目的である．本稿では金属研究に広く使われてい
る Embedded Atom Method(EAM)ポテンシャルを例に，物理的バックグラウンドと具体的な
フィッティング法について紹介する．また同時に，式の導出などの解法を通して，強力な武器と

2 原子間経験ポテンシャルの理論的バックグランド

図 1: Pair potentialと EAMの関数形状．

なる数式処理ソフトを紹介する．数式がやたらと出てくるが，使っている数学は高校の数 Iの範
囲を出ていないので，恐れずに一つ一つ追いかけていただきたい．
数式処理ソフトの使用法は紙面の都合で載せないが，拙著 [Nishitani01]が無料で手に入るの

でそちらを参照いただきたい．

2 原子間経験ポテンシャルの理論的バックグランド

2.1 二体ポテンシャルとEAMの表式

最初に二体ポテンシャルと EAMの表式を見ておこう．よく知られた二体間ポテンシャルには

Ei−atom =
∑

j

ψ(Rij) (1)

= A exp(−2pRij) − 2A exp(−pRij)

の形をしたMorse potentialがある．
単純な EAMポテンシャルは

Ei−atom =
∑

j

ψ(Rij) + F (ρi) (2)

=
∑

j

ψ(Rij) −
√
z |h(Rij)|

である．ここで 1行目の式が EAMの一般形を，2行目の式は完全結晶の場合の最も単純な形を
示している．ψ は単純な二体間ポテンシャル，F (ρ)は埋め込み関数 (embedding function)，ρ
は電子密度，z は配位数 (coordination number)である．相互作用は ψ と H を通して図 1のよ
うになっている．関数形はMorseの場合と同じであり，完全結晶では全く同じ振る舞いを示す．
このような形はどのようにして得られたのだろうか．また，これらのポテンシャルの相互作

用はどれもが exp 関数であるが，これにはどの程度の根拠があるのだろうか．原子間の結合は
電子軌道同士がどのような相互作用をしているかに立ち返って理解することが必要となる．

2.2 BOND

原子間の相互作用は２原子分子の化学結合を考えればその主な役者は出尽くす [Pettifor95,
Ch. 3]．原子 A,Bが結合を作る場合の Schödinger方程式を考えよう (図 2)．AB二量体の 1電
子 Schrödinger方程式を解きたい．すなわち

− h̄2

2m
∇2ψAB(r) + VAB(r)ψAB(r) = EABψAB(r) (3)

2 原子間経験ポテンシャルの理論的バックグランド 2.2 BOND

である．ここで VAB(r)は分子ポテンシャルである ．原理的にはこのポテンシャルはセルフコン
システントに決定されねばならない．つまり (3)式の左辺に入力として入れるポテンシャルは平
均電荷密度に依存し，したがって，(3)式の出力である電子波動関数 ψAB に依存する．
分子軌道法でよく使われる LCAO(Linear conbinations of atomic orbitals)の概念

ΨAB = cAΨA + cBΨB (4)

を使い，右側からΨA,ΨBを掛けて全空間で積分する．ここで二つの重要な積分関数Hamiltonian
matrix operatorと overlap matrix operator

Hα,β =

∫
ΨαĤΨβdr (5)

Sα,β =

∫
ΨαΨβdr (6)

を導入する．注意すべきは，これらの積分は経験的 tight binding理論では実際に積分すること
なく exp関数で近似される．ここで Ĥ は AB 二量体に対する Hamiltonian演算子で

Ĥ = − h̄2

2m
∇2 + VAB

である．LCAO永年方程式は(
HAA −E HAB − ESAB

HBA − ESBA HBB − E

)(
cA

cB

)
= 0 (7)

となる．すこしの変数の置き換え E = 1
2
(EA + EB) ,∆E = EB −EA によって，Hamiltonian

行列は (− 1
2
∆E −

(
E − E

)
h −

(
E − E

)
S

h −
(
E − E

)
S 1

2
∆E −

(
E − E

))(
cA

cB

)
= 0 (8)

のように変換される (厳密な導出はもう少し考察が必要．詳しくは [Pettifor95, Ch. 3]を参照せ
よ)．この行列式が意味のある解を持つ条件は永年方程式が０になる場合である．これはすぐに
解けて

E± = E +
|h|S ∓ 1

2

√
4h2 + (1 − S2)∆E2

1 − S2
(9)

となる．ここで (1 − S2)−1 = 1 + S2 + · · ·と展開すると
∼= E + |h|S ∓ 1

2

√
4h2 + ∆E2 (10)

という解を得る [課題 1]．これは二つの原子を近づけていった場合に新たに作られる結合準位と
反結合準位である (図 3)．ここで出てきた結果は EAMの形

Ei =
∑

ϕ−
√∑

h2 (11)

とほぼ一致している．

Free atom A Free atom B AB diatomic 
molecule

VA
EA

ΨA
+ =

VB
EB

ΨB
EA

EB

図 2: AB２原子分子のポテンシャルと電子軌道の模式図．

2 原子間経験ポテンシャルの理論的バックグランド 2.3 BAND

   wAB = 4h2 + ∆E2

図 3: AB２原子分子のエネルギー準位と電子分布．

図 4: 状態密度の矩形バンドモデル．

2.3 BAND

ここではもう一つの電子構造のとらえかたである，バンド理論から EAMの導出を見ておこ
う．Tight binding理論から原子の totalエネルギーを表すと

Ei−atom =
∑

j

ψ + Eeletron =
∑

j

ψ +

∫ EF

−∞
Eni(E)dE (12)

である．ここで ni(E)が電子の状態密度を，EF が Fermi準位を表す．この ni(E)を単純な矩
形で近似すると図 4のようになる．W をバンド幅とすると，バンドは −1/2W から 1/2W まで
広がっている．N0 がこのバンドに入る事ができるバンドの総数とすると，このバンドの高さは
N0/W で求まる．
一電子エネルギーは

Eeletron =

∫ EF

−W/2

E
N0

W
dE =

[
E2

2

N0

W

]EF

−W/2

(13)

となる．EF は N が実際にこの原子が持っている価電子数とすると

N =

∫ EF

−W/2

N0

W
dE =

[
E
N0

W

]EF

−W/2
(14)

で決められる．これを最後まで求めると

Eeletron = −1

2

WN(−N + N0)

N0
(15)

4



2 原子間経験ポテンシャルの理論的バックグランド 2.4 点欠陥生成エネルギー

となり，バンド幅W だけで電子エネルギーが決まっていることになる [課題 2]．バンド幅はバ
ンドのスペクトルを統計的に考える moment理論から導かれる．

moment は高校のとき習った統計分布の平均や分散を求める式を思い出していただきたい．
高次の momentは複雑な形状を再現するための指標のようなものである．p次の momentは

µp =
∑

n

εp
n =

∫ +∞

−∞
Epnα (E) dE =

∑
i1,i2···ip

Hi1,i2Hi2,i3 · · ·Hip,i1 (16)

と定義される．0次の momentは
µ0 = N (17)

より状態の総数を表し，1次の momentは

µ1 = Eα (18)

より Density of states(DOS)の重心を表す．また，規格化された 2次momentは

µ2/µ0 =
1

N

∑
j

H2
ij (19)

より平均次乗幅 (mean square width)を表す．これよりバンド幅は hopping integralH と配位
数 z とすると

√
zH2 で求まる．この結果も (2)式の EAMに一致する．

2.4 点欠陥生成エネルギー

このように EAM は，化学結合や電子バンドの描像の基礎的な記述から導くことが可能で，
その表式は電子の振る舞いを取り入れている．では，EAMと単純な２体間ポテンシャルとの違
いは何であるか?　その違いが端的に現われる空孔形成エネルギーを見てみよう．金属の点欠陥
生成エネルギー Ev と凝集 (cohesive) エネルギー Ec との関係を図 5 に示した．構造によらず
Ev = 3

8
∼ 1

4
Ec のほぼ直線的な関係が得られている．

点欠陥の生成エネルギーを考えるには図 6のようなモデルで考える必要がある．単一空孔は
一個の原子を抜きだして，それを完全結晶の中に埋め込むエネルギーに対応する．この操作の前
後で抜き出された原子の環境は変わらない．したがって，抜き出されたサイトの周りの原子がど
ういう変化を受けたかを計算すればよい．今簡単のために最近接原子数 (配位数)zのみに依存す
るポテンシャルを仮定する．そうすると vacancyの formation energyは

Ev = Efinal − Einitial = zE(z − 1) − zE(z) (20)

である．
Pair potentialの場合は

E(z) = −Ec = zψ0 (21)

であるから，
Ev = z(z − 1)ψ0 − z(z)ψ0 = −zψ0 = Ec = zψ (22)

となり，cohesive energyに等しいというよく知られた間違った答えが得られる．
EAMの場合も計算の仕方は同じで，

E(z) = zψ0 −√
z|h0| (23)

であるから，
Ev = z

{
(z − 1)ψ0 −√

z − 1|h0|
}
− z

{
(zψ0 −

√
z|h0|

}
(24)

となる．ここで相互作用に単純な exp 関数を仮定してその距離依存性を表す無次元対数微分
L = −dln(h)/dr を p, q として，Ec と r0 に fittingすると，

h0 = p/q
√
zψ0 (25)

ψ0 =
−Ec

(1 − p/q)z
(26)

となる．すこしの変形で

Ev =
2 − p/q

1 − p/q

Ec

2
(27)

3 フィッティング

Cohesive energy [eV]

E
v

 [
e

V
]

1086420
0

1

2

3

4

fcc

bcc

hcp

図 5: 点欠陥生成エネルギーと cohesiveエネルギーと
の関係．

perfect (initial) defect(final)

図 6: 点欠陥生成のモデル．

となる [課題 3]．金属でよく用いられている p/q = 3 ∼ 5を用いると前述の正しい関係が得られ
る [Allen76]．
このように正しい関係が得られた背景は以下のように理解されている．金属結合では電子が

結合間で共有されているという描像が描ける．vacancyのまわりで結合が一つ減ったとしても電
子は back bond側に流れ，その結合を強化する (back bond strengthening)．この変化を EAM
では

√
zの依存性として表現している．つまり，結合が完全結晶の配位数 z0から点欠陥の周りの

ように一つ減ったとしても，pair potentialのように線形にそのエネルギーが減少するのではな
く，

√
zで減少する．このようにして Ev と Ecの正しい関係が導かれる．このような back bond

strengtheningは界面や表面でも同様に起こっていると考えられる．

3 フィッティング
第一近接原子の相互作用だけを考えた EAMモデルへ，Bulk modulus(体積弾性率)，平衡原

子間距離，凝集エネルギーへポテンシャルをフィットさせる例を紹介する．また，このモデルを
用いて剪断定数を求める方法を示しておく．ここで示した表式は一般的であるので，エネルギー
計算部だけをより精度の高い第一原理計算などに置き換えることも可能である．

3.1 Bulk modulus

最初に EAMの表式．

6



3 フィッティング 3.1 Bulk modulus

broken bondvacancy

back bond

coordination number[z]

bo
nd

in
g 

en
er

gy   EC

  z0

 z

 z

broken bond
surface

back bond

図 7: Back bond problemの概念図とエネルギーの
√

z依存性．

> restart;
> E_total:=del->12*A*exp(-p*del)-B*sqrt(12*exp(-q*del)^2);

E total := del → 12Ae(−p del) −B
√

12 (e(−q del))2

ここでは平衡状態で規格化された距離 (r/r0 − 1)を使っている．その場合，距離の変
化率 (δ)をそのまま代入できる．次に対象とする (ここではAlの)物性値を入れる．

> BulkModulus:=77:
> Lat_const:=4.032:
> Ec:=3.39:
> Lat_number:=4:
> poq:=2.89:

先程示した欠陥生成エネルギー (0.8eV)から poqは 2.89としている．微小変化 (δ =
0.001)させたときの体積弾性率 (bulk modulus)に対応するエネルギー変化を eV–Å
単位で求めると，

> delta:=0.001:
> eVtoGPa:=160.218:
> Omega:=Lat_const^3/Lat_number:
> dE:=delta^2/2*9*Omega*BulkModulus/eVtoGPa;

dE := .00003543998357

となる．微小変形したときのエネルギーを求める．
> E1:=E_total(-0.001);
> E2:=E_total(-0.0);
> E3:=E_total(0.001);

E1 := 12Ae(.001 p) − 2B
√

3
√

(e(.001 q))2

E2 := 12. A− 3.464101615 B

E3 := 12Ae(−.001 p) − 2B
√

3
√

(e(−.001 q))2

まず E2から Aを求めておく．
> A:=solve(E2=-Ec,A);

A := .2886751346 B − .2825000000

3 フィッティング 3.2 Shear constants

qを引数として E1 = E3からBを求め，微小変形したときのエネルギー変化とBulk
Modulusから求めた正しいエネルギー変化との差を出力する関数を定義する．

> find_q:=proc(tmp)
> global B,p,q;
> B:=’B’;
> q:=tmp;
> p:=q*poq;
> B:=fsolve(E1=E3,B);
> evalf(E1-E2-dE);
> end;

q の上下限で関数が答えを返すかを確認しておく．
> q_min:=2:
> q_max:=5:
> plot(’find_q(x)’,x=q_min..q_max);

0

2e–05

4e–05

6e–05

8e–05

2.5 3 3.5 4 4.5 5
x

q を求める．
> q:=fsolve(’find_p(x)’,x,p_min..p_max);

q := 2.689779044

> A;B;p;q;

.1494688734

1.496384071565230481517682

7.773461438373904991942374

2.689779044

3.2 Shear constants

次に剪断定数を求める．最初に弾性定数の classical expressionと変形matrixの間の関係を
示しておく．

> restart;
> U/Omega=series(U(gamma1),gamma1=0,4);

U

Ω
= U(0) + D(U)(0) γ1 +

1

2
(D(2))(U)(0) γ12 +

1

6
(D(3))(U)(0) γ13 + O(γ14)

ここで一次の項は平衡条件より消え，3次以上の項は高次項として無視すると，二次
の項だけが残る．弾性定数の classical expressionは

> U:=1/2*C_11*(ex^2+ey^2+ez^2)+
> C_12*(ex*ey+ey*ez+ez*ex)+
> C_44*(1/2*exy^2+1/2*eyz^2+1/2*ezx^2);

U1 :=
1

2
C 11 (ex 2 + ey2 + ez 2) + C 12 (ex ey + ey ez + ez ex )

+ C 44 (
1

2
exy2 +

1

2
eyz 2 +

1

2
ezx2)

U1は体積 (Ω)で割ってあるとする．eeを変形 matrixと考えると

8



3 フィッティング 3.2 Shear constants

> U1:=subs({ex=ee[1,1],ey=ee[2,2],ez=ee[3,3],
> exy=ee[1,2]+ee[2,1],eyz=ee[2,3]+ee[3,2],ezx=ee[1,3]+ee[3,1]},U1);

U1 :=
1

2
C 11 (ee1, 1

2 + ee2, 2
2 + ee3, 3

2) + C 12 (ee1, 1 ee2, 2 + ee2, 2 ee3, 3 + ee3, 3 ee1, 1)

+C 44 (
1

2
(ee1, 2 + ee2, 1)

2 +
1

2
(ee2, 3 + ee3, 2)

2 +
1

2
(ee1, 3 + ee3, 1)

2)

となる．C’の変形 matrixは
> CprimeM:=[[-delta+delta*delta,0,0],
> [0,-delta+delta*delta,0],
> [0,0,2*delta+delta*delta]];
> ee:=matrix(CprimeM);

CprimeM := [[−δ + δ2, 0, 0], [0, −δ + δ2, 0], [0, 0, 2 δ + δ2]]

ee :=

[ −δ + δ2 0 0
0 −δ + δ2 0
0 0 2 δ + δ2

]
である．これを微分して無限小まで持っていくと，この変形 matrixによって C′ =
1/2(C11 − C12) の変形に対応するエネルギー変化が求まることが分かる．

> limit(diff(U1,delta,delta),delta=0)/6/2;

1

2
C 11 − 1

2
C 12

次に先ほど求めた EAMポテンシャルでの剪断定数を求める．

原子配置を作る．
> with(combinat, permute):
> NN_list:=[]:
> seed:=[[1/2,1/2,0],[1/2,-1/2,0],[-1/2,-1/2,0]]:
> for i from 1 to 3 do
> NN_list:=[op(NN_list),op(permute(seed[i]))];
> od:
> NN_list;

[[
1

2
,
1

2
, 0], [

1

2
, 0,

1

2
], [0,

1

2
,
1

2
], [

1

2
,
−1

2
, 0], [

1

2
, 0,

−1

2
], [

−1

2
,
1

2
, 0], [

−1

2
, 0,

1

2
], [0,

1

2
,
−1

2
], [0,

−1

2
,
1

2
],

[
−1

2
,
−1

2
, 0], [

−1

2
, 0,

−1

2
], [0,

−1

2
,
−1

2
]]

いくつかの関数 (distance, rep(r), hop(r))を定義．
> sigma:=evalf(1/sqrt(2)):
> distance:=proc(vec1)
> evalf(sqrt(add(vec1[i]*vec1[i],i=1..3)));end:

> rep:=proc(dist)
> global A,sigma,p;
> local r;
> r:=dist/sigma;
> A*exp(-p*(r-1.0));
> end:
> hop:=proc(dist)
> global B,sigma,q;
> local r;
> r:=dist/sigma;
> B*exp(-q*(r-1.0));
> end:

周囲の原子座標を引数としてエネルギーを計算する関数を定義．

3 フィッティング 3.2 Shear constants

> total_energy:=proc(N_list)
> local energy,rho,n_nei,i,dist,hopping;
> energy:=0;
> rho:=0;
> n_nei:=nops(N_list);
> for i from 1 to n_nei do
> dist:=distance(N_list[i]);
> energy:=energy+rep(dist);
> hopping := hop(dist);
> rho := rho + hopping*hopping;
> od;
> energy := energy - sqrt(rho);
> end:

変形マトリックスを定義 [Alinaghian94]．
> delta:=’delta’:
> with(LinearAlgebra):
> BulkM:=[[delta,0,0],[0,delta,0],[0,0,delta]]:
> C11M:=[[delta,0,0],[0,0,0],[0,0,0]]:
> C44M:=[[delta*delta,delta,delta],
> [delta,delta*delta,delta],
> [delta,delta,delta*delta]]:
> CprimeM:=[[-delta+delta*delta,0,0],
> [0,-delta+delta*delta,0],
> [0,0,2.0*delta+delta*delta]]:
> strain_parameter:=[[2/9,BulkM,"BulkModulus"],
> [2.0,C11M,"C11"],[1/6,C44M,"C44"],[1/6,CprimeM,"C’"]]:

変形 matrixにしたがってリストの座標を変位させる関数を定義．
> with(LinearAlgebra):
> Mk_strained:=proc(M_strain,N_list)
> local StrainM,Strained_list,n_nei,i:
> StrainM:=Matrix(M_strain)+Matrix(3,3,shape=identity);
> Strained_list:=[];
> n_nei:=nops(N_list);
> for i from 1 to n_nei do
> Strained_list:=[op(Strained_list),convert(Multiply(StrainM,Vector(N_li
> st[i])),list)];
> od:
> Strained_list;
> end:

> with(stats):

変位位置を入力にして，エネルギーを計算し，2 次関数にフィットして剪断定数を
計算．

> for strain_index from 1 to 4 do
> x1:=[];
> e1:=[];
> for i from -2 to 2 do
> delta:=evalf(i*0.0025);
> x1:=[op(x1),delta];
> E_s:=total_energy(Mk_strained(strain_parameter[strain_index][2],NN_lis
> t));
> e1:=[op(e1),E_s];
> od:
> data:=[x1,e1]:
> b:=’b’;x:=’x’;
> fit1:=fit[leastsquare[[x,y], y=a+b*x+c*x^2, {a,b,c}]](data);
> b:=coeff(rhs(fit1),x^2);
> shear_res:=strain_parameter[strain_index,1]/Omega*b*eVtoGPa;
> printf("%12s [GPa] =
> %6.2g\n",strain_parameter[strain_index,3],shear_res);
> od:

BulkModulus [GPa] = 77.02
C11 [GPa] = 95.14
C44 [GPa] = 27.19
C’ [GPa] = 13.6

4 まとめ
量子力学の単純なモデルから経験的ポテンシャルがどのように導かれるか，そのポテンシャ

ルを現実の結晶にどのようにフィットさせるかを簡単に示した．点欠陥生成エネルギーや粒界エ
ネルギーはここで示した単純な EAMポテンシャルでも十分に再現することが可能である．しか
し，より微妙な結合の変化を表現するには，結合の環境依存性を取り入れる必要がある．精度の
高い第一原理計算結果や，より複雑な分子動力学，モンテカルロシミュレーションで使われるモ
デルの仮定や精度を理解してパソコンを材料開発に積極的に利用していただきたい．

Mapleはここで主に紹介した数式処理，数値計算などのほかにもデータの三次元表示など優
れた機能を持っている．また，多くのプラットフォーム上で同じソフトが使えるのも魅力である．
Unix上では以下の例で示すような filterとしての使用も可能で，定形処理を大幅に簡略化して
くれる．多くの研究者が数式処理ソフトを，数学を扱う強力な道具として積極的に使われるよう
願います．

List 4.1 Unix上で input.dataからデータを読み output.resへ出力するスクリプト (sample.mws)．
maple ＜ sample.mwsと使う．
interface(quiet=true);
with(stats):with(linalg):
data:=readdata("input.data",4):

... <処理を記述> ...

writeto("output.res"):
printf("Integrated Free Energy: %15.10f\n",result):
writeto(terminal):
interface(quiet=false):

引用文献
[Alinaghian94] P. Alinaghian, S.R. Nishitani, and D.G. Pettifor, Phil. Mag. B, 69(1994),

889.

[Allen76] G.Allen and M.Lannoo, J.Phys.Chem.Solids, 37(1976), 699.

[Nishitani01] 西谷滋人, 利用の手引き-Maple 編-,(京都大学大型計算機センター 2001).
http://www.mtl.kyoto-u.ac.jp/users/bob/Maple/Maple.html.

[Pettifor95] D.G. Pettifor, Bonding and Structure of Molecules and Solids, (1995 Clarendon
Press, Oxford). 分子・固体の結合と構造, (1997 技報堂).

参考文献
一般的な参考文献としては，引用文献 Pettifor95以外に以下のものがある．　

[1] Ed by D.G. Pettifor and A.H. Cottrell, Electron Theory in Alloy Design,(1992, The
Institute of Materials, London).

[2] R. Philips, Crystals, Defects and Microstructures –Modeling Across Scales–,(2001,
Cambridge University Press, Cambridge).

以下は review．

[3] A.E. Carlsson, Beyond Pair Potentials in Elemental Transition Metals and Semicon-
ductors,Solid State Phys., 43(1990), 1.　（欠陥構造のシミュレーションに適用できる
ポテンシャルの要件を議論）

[4] M.S.Daw, S.M.Foiles, and M.I.Baskes, The embedded-atom method: a review of theory
and applications, Mater. Sci. Rep. 9(1993), 251-310.（EAM全般）

[5] Ed by A.F. Voter, Interatomic Poetntials for Atomistic Simulations, MRS Bulletin.
(February 1996), 17–48. （EAM他の最新のポテンシャルの応用例を紹介）

[6] S. Erkoc, Empirical potential energy functions used inthe simulations omateriasl prop-
erties, Annual Reviews of Computational Physics, IX, Ed. by D. Stauffer, 2001 World
Scientific Pub. Comp., pp.1–103.　（90もの経験ポテンシャルの数式と文献がまとめら
れている）

課題
課題 1： (7)式から (10)式を導出せよ．

[解答例]：先ず，Hamiltonianの作る．
> with(linalg):

> diago1:=-1/2*DE-(E1-Eav);
> diago2:=1/2*DE-(E1-Eav);
> offdiago:=h-(E1-Eav)*S;

diago1 := −1

2
DE − E1 + Eav

diago2 :=
1

2
DE − E1 + Eav

offdiago := h − (E1 − Eav)S

> ham1:=matrix(2,2,[diago1,offdiago,offdiago,diago2]);

ham1 :=

 −1

2
DE − E1 + Eav h− (E1 − Eav )S

h− (E1 − Eav)S
1

2
DE − E1 + Eav


次に，永年方程式を書きだし，解を求める．

> sol1:=det(ham1);

sol1 := −1

4
DE2 + E1 2 − 2E1 Eav + Eav2 − h2 + 2hS E1 − 2hS Eav − S2 E1 2

+ 2S2 E1 Eav − S2 Eav2

> solve(sol1=0,E1);

1

2

8hS − 8Eav + 8S2 Eav + 4
√

DE2 + 4 h2 − S2 DE2

−4 + 4S2
,

1

2

8hS − 8Eav + 8S2 Eav − 4
√

DE2 + 4 h2 − S2 DE2

−4 + 4S2

bonding levelを取りだして，S の 2次までの級数展開を求める．

> Eplus:=%[1];

Eplus :=
1

2

8hS − 8Eav + 8S2 Eav + 4
√

DE2 + 4h2 − S2 DE2

−4 + 4S2

> series(Eplus,S,2);

(Eav − 1

2

√
DE2 + 4h2) − hS + O(S2)

課題 2： (15)式を求めよ．(解答略)

課題 3： (27)式を導出せよ．

[解答例]

> restart;

> E_t:=z->z*p0-sqrt(z)*H0;

E t := z → z p0 −√
zH0

> Ev:=z*E_t(z-1)-z*E_t(z);

Ev := z ((z − 1) p0 −√
z − 1H0 )− z (z p0 −√

zH0 )

> H0:=poq*sqrt(z)*p0;

H0 := poq
√
z p0

> p0:=-Ec/(1-poq)/z;

p0 := − Ec

(1 − poq) z

> assume(z>1);
> series(subs(z=1/x+1,normal(Ev)),x,1);

−
Ec (1 − 1

2
poq)

−1 + poq
+ O(

√
x)

課題 4： 3 節の fitting script に手を加え q を一定値に固定した場合の fittingをおこなえ．(解
答略)

課題 5：剪断定数の q 値依存性を議論せよ．(解答略)

課題 6： Alのポテンシャルを使って，未緩和状態での (111)表面エネルギーを求めよ．(解答略)

